
1 www.handsontec.com 

 

 

Handson Technology 

Data Specs 

DS18B20+ Digital Temperature Sensor 
 

This is a pre-wired and waterproofed version of the DS18B20 sensor. Handy for when you need to measure 

something far away, or in wet conditions. While the sensor is good up to 125°C the cable is jacketed in PVC so 

we suggest keeping it under 100°C. Because they are digital, you don't get any signal degradation even over 

long distances! These 1-wire digital temperature sensors are fairly precise (±0.5°C over much of the range) and 

can give up to 12 bits of precision from the onboard digital-to-analog converter. They work great with any 

microcontroller using a single digital pin, and you can even connect multiple ones to the same pin, each one has 

a unique 64-bit ID burned in at the factory to differentiate them. Usable with 3.0-5.0V systems. 

 

 

 

 

 

SKU: SSR1001 

 

Brief Data: 

 Usable temperature range: -55 to 125°C (-67°F to +257°F) 

 9 to 12 bit selectable resolution. 

 Uses 1-Wire interface- requires only one digital pin for communication. 

 Unique 64 bit ID burned into chip. 

 Multiple sensors can share one pin. 

 ±0.5°C Accuracy from -10°C to +85°C. 

 Temperature-limit alarm system. 

 Query time is less than 750ms. 

 Usable with 3.0V to 5.5V power/data. 

 Waterproof. 

 

http://handsontec.com/index.php/product/ds18b20-digital-temperature-sensor/

Connection Pins Assignment: 

 

 

 

Mechanical Dimension: 

Unit: mm 

Application Examples with Arduino: 

Connect up the sensor to Arduino Uno board as shown in below schematic: 

Add a 4.7KΩ resistor between power supply pin and signal pin as the digital pin is open drain type which need 

a pull up voltage. 

 

Arduino Sketch: 

You need an Arduino Library for One-Wire communication with Arduino, download the library from the below 

link and extract it to Arduino IDE, “Arduino>libraries” folder. 

www.handsontec.com/dataspecs/Arduino_Lib/OneWire-master.zip 

 

 

 

 

Open up the Arduino IDE. Go to the examples file and double click “DS18x20_Temperature” sketch. 

 

http://www.handsontec.com/dataspecs/Arduino_Lib/OneWire-master.zip

Arduino Sketch DS18x20_Temperature: 

 

#include <OneWire.h> 

 

// OneWire DS18S20, DS18B20, DS1822 Temperature Example 

// 

// http://www.handsontec.com 

 

OneWire  ds(10);  // on pin 10 (a 4.7K resistor is necessary) 

 

void setup(void) { 

  Serial.begin(9600); 

} 

 

void loop(void) { 

  byte i; 

  byte present = 0; 

  byte type_s; 

  byte data[12]; 

  byte addr[8]; 

  float celsius, fahrenheit; 

   

  if ( !ds.search(addr)) { 

    Serial.println("No more addresses."); 

    Serial.println(); 

    ds.reset_search(); 

    delay(250); 

    return; 

  } 

   

  Serial.print("ROM ="); 

  for( i = 0; i < 8; i++) { 

    Serial.write(' '); 

    Serial.print(addr[i], HEX); 

  } 

 

  if (OneWire::crc8(addr, 7) != addr[7]) { 

      Serial.println("CRC is not valid!"); 

      return; 

  } 

  Serial.println(); 

  

  // the first ROM byte indicates which chip 

  switch (addr[0]) { 

    case 0x10: 

      Serial.println("  Chip = DS18S20");  // or old DS1820 

      type_s = 1; 

      break; 

    case 0x28: 

      Serial.println("  Chip = DS18B20"); 

      type_s = 0; 

      break; 

    case 0x22: 

      Serial.println("  Chip = DS1822"); 

      type_s = 0; 

      break; 

    default: 

      Serial.println("Device is not a DS18x20 family device."); 

      return; 

  }  

 

  ds.reset(); 

  ds.select(addr); 

  ds.write(0x44, 1);        // start conversion, with parasite power on at the end 

   

  delay(1000);     // maybe 750ms is enough, maybe not 

  // we might do a ds.depower() here, but the reset will take care of it. 

   

  present = ds.reset(); 

  ds.select(addr);     

  ds.write(0xBE);         // Read Scratchpad 

 

  Serial.print("  Data = "); 

  Serial.print(present, HEX); 

  Serial.print(" "); 

  for ( i = 0; i < 9; i++) {           // we need 9 bytes 

    data[i] = ds.read(); 

    Serial.print(data[i], HEX); 

    Serial.print(" "); 

  } 

  Serial.print(" CRC="); 

  Serial.print(OneWire::crc8(data, 8), HEX); 

  Serial.println(); 

 

  // Convert the data to actual temperature 

  // because the result is a 16 bit signed integer, it should 

  // be stored to an "int16_t" type, which is always 16 bits 

  // even when compiled on a 32 bit processor. 

  int16_t raw = (data[1] << 8) | data[0]; 

  if (type_s) { 

    raw = raw << 3; // 9 bit resolution default 

    if (data[7] == 0x10) { 

      // "count remain" gives full 12 bit resolution 

      raw = (raw & 0xFFF0) + 12 - data[6]; 

    } 

  } else { 

    byte cfg = (data[4] & 0x60); 

    // at lower res, the low bits are undefined, so let's zero them 

    if (cfg == 0x00) raw = raw & ~7;  // 9 bit resolution, 93.75 ms 

    else if (cfg == 0x20) raw = raw & ~3; // 10 bit res, 187.5 ms 

    else if (cfg == 0x40) raw = raw & ~1; // 11 bit res, 375 ms 

    //// default is 12 bit resolution, 750 ms conversion time 

  } 

  celsius = (float)raw / 16.0; 

  fahrenheit = celsius * 1.8 + 32.0; 

  Serial.print("  Temperature = "); 

  Serial.print(celsius); 

  Serial.print(" Celsius, "); 

  Serial.print(fahrenheit); 

  Serial.println(" Fahrenheit"); 

} 

 

Upload the sketch into Arduino Uno board and open up the Serial Monitor, the below windows shown the 

sensor being detected and displays the measure temperature: 

 

 

 

Web Resources: 

 OneWire Library: http://www.pjrc.com/teensy/td_libs_OneWire.html 

  

http://www.pjrc.com/teensy/td_libs_OneWire.html

Handsontec.com 

We have the parts for your ideas 
 

HandsOn Technology provides a multimedia and interactive platform for 

everyone interested in electronics. From beginner to diehard, from student to 

lecturer. Information, education, inspiration and entertainment. Analog and 

digital, practical and theoretical; software and hardware.  

 

 

 

HandsOn Technology support Open Source Hardware (OSHW) 
Development Platform. 
 

 

 

 

 

 

 

Learn : Design : Share 
 

 

handsontec.com 
 

 

 

http://www.handsontec.com/
https://handsontec.com/

The Face behind our product quality… 

In a world of constant change and continuous technological development, a new or replacement product 

is never far away – and they all need to be tested. 

Many vendors simply import and sell wihtout checks and this cannot be the ultimate interests of anyone, 

particularly the customer. Every part sell on Handsotec is fully tested. So when buying from Handsontec 

products range, you can be confident you’re getting outstanding quality and value. 

 

We keep adding the new parts so that you can get rolling on your next project. 

 
Breakout Boards & Modules 

 
Connectors Electro-Mechanical Parts 

 
Engineering Material 

 
Mechanical Hardware 

 

P 

Electronics Components 

Power Supply Arduino Board & Shield 
 

Tools & Accessory 

 

 

http://handsontec.com/index.php/product-category/breakout-boards/
http://handsontec.com/index.php/product-category/connectors/
http://handsontec.com/index.php/product-category/electro-mechanical/
http://handsontec.com/index.php/product-category/engineering-material/
http://handsontec.com/index.php/product-category/mechanical-parts/
http://handsontec.com/index.php/product-category/electronics-parts/
http://handsontec.com/index.php/product-category/power-supply/
http://handsontec.com/index.php/product-category/arduino-boards/
http://handsontec.com/index.php/product-category/tools-accessories/
http://handsontec.com/index.php/product-category/breakout-boards/
http://handsontec.com/index.php/product-category/connectors/
http://handsontec.com/index.php/product-category/electro-mechanical/
http://handsontec.com/index.php/product-category/engineering-material/
http://handsontec.com/index.php/product-category/mechanical-parts/
http://handsontec.com/index.php/product-category/electronics-parts/
http://handsontec.com/index.php/product-category/power-supply/
http://handsontec.com/index.php/product-category/arduino-boards/
http://handsontec.com/index.php/product-category/tools-accessories/

